
Simulation

This lecture describes how to use Verilog to simulate designs.
After this lecture you should be able to write a testbench that can: set initial values, generate clocks, read test vectors from
a file, display values, and terminate on a condition.
Version 2: corrected testbench.

Simulation

Verilog can be used to test HDLdesigns by simulating
their operation. A simulation consists of the module
being tested (called the Design Under Test or DUT)
that is instantiated in another module called a test-
bench. The testbench applies inputs to the DUT and
checks its outputs:

Test Vectors

The inputs to the DUT and the corresponding ex-
pected outputs are called test vectors. These can be
generated by the testbench itself or they can be read
from a file.
Test vectors should include:

1. typical inputs,
2. minimum and maximum valid inputs,
3. invalid inputs, and
4. randomly-chosen values.

Exercise 1: Give examples of appropriate test inputs for each of
the above categories if youwere testing a circuit that computed the
square root of a 16‑bit signed number.

Verilog for Verification

The following Verilog features are useful for simula-
tion but can’t be synthesized (implemented in hard-
ware).

initial and always blocks

initial blocks execute once at the start of the sim-
ulation and are used to initialise signals. always
blocks execute continuously. begin and end are used
to group statements in these blocks (as with { and }
in C).

if/else/for/while

if/else, for, and while statements, whose syntax
is similar to C, can be used in initial or always
blocks. However, the ability to execute statements
based on conditions or events reduces the need for
these statements.

Delays

Placing #number before a statement delays1 execu-
tion by number simulation time. The suffixes ns and
us can be used for nano- and micro-seconds.
The syntax @(event) where event can be posedge

or negedge before a signal name or just a signal name
delays execution until that signal edge or a change in
that signal value.

wait

The wait(expression) statement pauses until the ex-
pression is non-zero.
Exercise 2: What’s the difference between wait(x) y='1; and
@(x) y='1;?

System Tasks

Functions beginning with $ are called system tasks.
Useful ones include:

1Delays are not synthesizable because delays cannot be easily
implemented in hardware.

Figure 1: Simulation waveforms (from ex66.vcd).

• $display() similar to C’s printf(), can be
used to print values during a simulation;

• $dumpfile and $dumpvars record changes in
signals to a .vcd file for subsequent viewing
with a waveform viewer.

• $fopen() and $fscanf(), similar to the C li-
brary functions fopen() and fscanf(), can
open and read from text files.

• $finish and $stop terminate or suspend a sim-
ulation.

Example

The testbench below demonstrates the language fea-
tures described above.
The DUT is a module with an 8-bit input that out-

puts a 16-bit sum of all the odd-valued inputs since a
reset input was asserted. An initial block opens a
file containing test vectors, initializes the clock signal
and $stop’s the simulation at the end of the test vec-
tor file. Every 0.5 µs the clock is inverted; this creates
a 1MHz clock.
The final always block continuously reads and ap-

plies the input(s) from each test vector, waits for the
falling edge of the clock, and compare the DUT out-
put(s) to the value(s) read from the test vectors:

initial block executes here
inputs applied

outputs checked

time

(negedge clk)

// output cumulative sum of odd-valued inputs

module ex66
( input logic reset, clk,
input logic [7:0] in,
output logic [15:0] out ) ;

always_ff @(posedge clk)
out <= reset ? '0 :

in & 1 ? out+in : out ;

endmodule

// example testbench

module ex66_tb ;

// DUT inputs and outputs
logic reset, clk ;
logic [7:0] in ;
logic [15:0] out, out_ ;

// file descriptor and values read per fscanf()
integer fd, n ;

// instantiate DUT
ex66 ex66_0 (.*) ;

// initialization
initial begin

// record all signals in a .vcd file
$dumpfile("ex66.vcd") ;
$dumpvars ;

// initialize clock
clk = '0 ;

// open the test vector file
fd = $fopen("ex66data.csv","r") ;

// wait for end of file or error
wait (n < 0) $stop ;

end

// generate 1 MHz clock (2x500ns period)
always #0.5us clk = ∼clk ;

// read/apply/check test vectors
always begin

// read & set DUT inputs; display test vector
n = $fscanf(fd,"%d,%d,%d", reset, in, out_) ;
$display ("test vector=%d,%d,%d",reset,in,out_) ;

// on falling clock edge
@(negedge clk) begin

// display DUT internal signals; check outputs
$display ( "DUT state = %d", ex66_0.out ) ;
if ( out != out_ ) begin
$display("***Error: %d, %d, %d",in,out,out_);

end

end
end

endmodule

The test vectors are read from the file
ex66data.csv containing the following lines2:

1,0,0
0,1,1
0,2,1
0,3,4
0,4,4
0,5,8
1,7,0

Running this testbench using the Modelsim simu-
lation program creates the waveform files shown in
Figure 1 and, since the test vector contains an error,
the following line is printed showing the input, ac-
tual output and the expected output given in the test
vector:

# test vector=1, 0, 0
# DUT state = 0
# test vector=0, 1, 1
# DUT state = 1
# test vector=0, 2, 1
# DUT state = 1
# test vector=0, 3, 4
# DUT state = 4
# test vector=0, 4, 4
# DUT state = 4
# test vector=0, 5, 8
# DUT state = 9
# ***Error: 5, 9, 8
# test vector=1, 7, 0
# DUT state = 0
# test vector=1, 7, 0

Exercise 3: How could you:

(a) terminate the simulation if a test vector failed?
(b) change the clock frequency to 10 MHz?
(c) print each test vector as it’s read?
(d) assert the reset input for two clock cycles?

2This file is in Comma Separated Values (.csv) format.

